



Metrum Research Group LLC  
Phone: 860.735.7043  
charlesm@metrumrg.com

2 Tunxis Road, Suite 112  
Tariffville, CT 06081  
metrumrg.com

# Torsten: A Prototype Model Library for Bayesian PKPD Modeling in Stan

## User Manual: Version 0.81

Charles Margossian and Bill Gillespie

## DEVELOPMENT

This project is open-source. We have received extensive help and advice from Stan's development team, as well as insightful feedbacks from the broader Stan community. We are active contributors to Stan's core language, and our focus is on tools for differential equations based models, with the goal to support applications in pharmacometrics. Metrum Research Group's collaboration with Columbia University for the development of tools for *Fast and Flexible Differential Equation Model Fitting with Application to Pharmacometrics* is supported by the Small Business Technology Transfer grant from the Office of Naval Research.

**Licensing.** Torsten is licensed under the BSD 3-clause license.

## 1. INTRODUCTION

Metrum Research Group has developed a prototype Pharmacokinetic/Pharmacodynamic (PKPD) model library for use in Stan 2.12. The current version includes:

- Specific linear compartmental models:
  - One compartment model with first order absorption
  - Two compartment model with elimination from and first order absorption into central compartment
- General linear compartmental model described by a matrix exponential
- General compartmental model described by a system of first order Ordinary Differential Equations (ODEs)

The models and data format are based on NONMEM®<sup>1</sup>/NMTRAN/PREDPP conventions including:

- Recursive calculation of model predictions
  - This permits piecewise constant covariate values
- Bolus or constant rate inputs into any compartment
- Handles single dose and multiple dose histories
- Handles steady-state dosing histories for specific and general linear models
- Implemented NMTRAN data items include: TIME, EVID, CMT, AMT, RATE, ADDL, II, SS

This library provides Stan language functions that calculate amounts in each compartment.

### Implementation details.

- Stan (<http://mc-stan.org/>) [1]
- All functions are programmed in C++ and are compatible with the Stan-math autodiff library
- All functions can be called directly in a Stan file in a manner identical to other built-in functions
- One and two compartment models: coded analytical calculations
- General linear compartment models with semi-analytical solutions using built-in (still in development) Matrix Exponential function (Pade approximation coupled with scaling and squaring [2])
- General compartment models with numerical solutions to ODEs using built-in ODE integrators in Stan (Runge-Kutta 4th/5th and CVODES), with adjustable tuning parameters

**WARNING:** The current version of Torsten is a *prototype*. It is being released for review and comment, and to support limited research applications. It has not been rigorously tested and should not be used for critical applications without further testing or cross-checking by comparison with other methods.

**Development plans.** Our current plans for future development of Torsten include the following:

- A system to easily share packages of Stan functions (written in C++ or in the Stan language)
- Extend formal tests
  - C++ Google unit tests
  - Comparison with simulations from the R package *mrgsolve*
- Steady state calculation for the General compartment model
  - This will require the developer of a solver for nonlinear algebraic equations (aka root solver)
- Allow user to provide time-varying constant rate matrix in `linCptModel`

---

<sup>1</sup>NONMEM® is licensed and distributed by ICON Development Solutions.

- Optimize Matrix exponential functions
  - Function for action of Matrix Exponential on a vector
  - Coded gradients
  - Special algorithm for matrices with special properties
- Google C++ unit tests
- R tests comparing simulations with the R package *mrgsolve*
- Conduct beta testing

## 2. INSTALL STAN AND TORSTEN

Installation files are available on GitHub: <https://github.com/charlesm93/example-models/tree/feature/issue-70-PKPDexamples-torsten/PKPD/torsten>

Torsten uses a development version of Stan, that follows the 2.12 release, in order to implement the matrix exponential function required for the general linear compartment model. The easiest way to install Stan with the CmdStan interface is to run the bash file `setupTorsten.sh`. CmdStan-dev is set to the last version Torsten was tested on.

Download CmdStan:

```
git clone https://github.com/stan-dev/cmdstan.git
cd cmdstan
git reset --hard 53b4041
```

Download Stan with Torsten:

```
git clone https://github.com/charlesm93/stan.git
```

Download Stan\_math with Torsten:

```
cd stan
cd lib
rm -r stan_math
git clone https://github.com/charlesm93/math.git
mv math stan_math
```

### 3. USING TORSTEN

In this section we go through the different functions Torsten adds to Stan. It will be helpful to apply these functions to a simple example.

**Example 1: Two Compartment Model.** We modeled drug absorption in a single patient and simulated plasma drug concentrations:

- Single Dose: 5000 mg
- PK: plasma concentration of parent drug ( $c$ )
- PK measured at 0, 0.25, 0.50, 0.75, 1, 1.25, 1.50, 1.75, 2, 4, 6, 8, 10, 12, 14, 16, 18, and 20 hours after dosing

The plasma concentration ( $c$ ) were simulated according to the following equations:

$$\begin{aligned}\log(c) &\sim N(\log(\hat{c}), \sigma^2) \\ \hat{c} &= f_{2cpt}(t, CL, Q, V_2, V_3, k_a) \\ (CL, Q, V_2, V_3, k_a) &= (5 \text{ L/h}, 8 \text{ L/h}, 20 \text{ L}, 70 \text{ L}, 1.2 \text{ h}^{-1}) \\ \sigma &= 0.05\end{aligned}$$

The data are generated using the R package *mrgsolve*, see `TwoCptModelSimulation.R`. We show the results obtained when using the function `PKModelTwoCpt`, which computes solutions to the ODEs analytically.

**Linear One and Two Compartment Model.** The one and two compartment model functions have the form:

```
<model name>(theta, time, amt, rate, ii, evid, cmt, addl, ss)
```

where `time`, `amt`, `rate`, `ii` are arrays of real and `evid`, `cmt`, `addl`, `ss` arrays of integers. All arrays have the same length, which corresponds to the number of events. `theta`, also known as the *parameter matrix*, is an array of vectors. If the parameters are constant for all events, `theta` should be an array of length 1, else its length should be the number of events. In the latter case, the  $i$ th row contains a vector parameters for the time interval `[time[i-1], time[i]]`.

The options for *model name* are:

- `PKModelOneCpt`
- `PKModelTwoCpt`

which respectively correspond to the one and two compartment model with first order absorption (figure 1). A vector in `theta` is expected to contain parameters  $CL$ ,  $V_2$ , and  $k_a$  for the one compartment case, and  $CL$ ,  $Q$ ,  $V_2$ ,  $V_3$ , and  $k_a$  for the two compartments case, [in this order](#), followed by the bioavailability fraction of each compartment (non-effective if set to 1) and the lag time in each compartment (non-effective if set to 0). Note that setting  $k_a$  to 0 eliminates the first-order absorption.




FIGURE 1. One and two compartment models with first order absorption implemented in Torsten.

`PKModelTwoCpt` can be used to fit example 1, see `TwoCptModelExample.stan`. Four MCMC chains of 2000 iterations were simulated. The first 1000 iteration of each chain were discarded. Thus 1000 MCMC samples were used for the subsequent analyses.

**Result.** The MCMC history plots (figure 3) suggest that the 4 chains have converged to common distributions for all of the key model parameters. The fit to the plasma concentration data (figure 5) are in close agreement with the data, which is not surprising since the fitted model is identical to the one used to simulate the data. Similarly the parameter estimates summarized in Table 1 are consistent with the values used for simulation.

TABLE 1. Summary of the MCMC simulations of the marginal posterior distributions of the model parameters

	mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
CL	5.19	0.01	0.25	4.60	5.05	5.21	5.35	5.62	971.04	1.00
Q	7.90	0.01	0.47	6.97	7.58	7.93	8.21	8.79	1662.02	1.00
V2	20.25	0.08	2.27	15.71	18.74	20.23	21.76	24.67	896.67	1.00
V3	64.10	0.22	6.98	52.60	59.48	63.38	67.74	80.65	1032.66	1.00
ka	1.25	0.00	0.17	0.94	1.13	1.24	1.35	1.62	940.99	1.00

**General Linear Compartment Model.** A general linear compartment model refers to a model that may be described in terms of a system of first order linear differential equations with (piecewise) constant coefficients, i.e., a differential equation of the form:

$$x'(t) = Kx(t)$$

FIGURE 2. Stan language for fitting a two compartment model using the PKModelTwoCpt function (abstract)

```

data {
  int<lower = 1> nt; # number of events
  int<lower = 1> nObs; # number of observation
  int<lower = 1> iObs[nObs]; # index of observation
  int<lower = 1> cmt[nt];
  int evid[nt];
  int addl[nt];
  int ss[nt];
  real amt[nt];
  real time[nt];
  real rate[nt];
  real ii[nt];

  vector<lower = 0>[nObs] cObs; # observed concentration (Dependent Variable)
}

      :

parameters {
  real<lower = 0> CL;
  real<lower = 0> Q;
  real<lower = 0> V2;
  real<lower = 0> V3;
  real<lower = 0> ka;
  real<lower = 0> sigma;
}

      :

transformed parameters {

  theta[1][1] = CL;
  theta[1][2] = Q;
  theta[1][3] = V2;
  theta[1][4] = V3;
  theta[1][5] = ka;
  theta[1][6] = 1; # F1
  theta[1][7] = 1; # F2
  theta[1][8] = 1; # F3
  theta[1][9] = 0; # tlag1
  theta[1][10] = 0; # tlag2
  theta[1][11] = 0; # tlag3

  x = PKModelTwoCpt(theta, time, amt, rate, ii, evid, cmt, addl, ss);

  cHat = col(x, 2) ./ V2; # get concentration in the central compartment

  for(i in 1:nObs){
    cHatObs[i] = cHat[iObs[i]]; # predictions for observed data records
  }
}

model {
  # priors
  CL ~ lognormal(log(10), 0.25);
  Q ~ lognormal(log(15), 0.5);
  V2 ~ lognormal(log(35), 0.25);
  V3 ~ lognormal(log(105), 0.5);
  ka ~ lognormal(log(2.5), 1);
  sigma ~ cauchy(0, 1);

  logCObs ~ normal(log(cHatObs), sigma);
}

```






FIGURE 3. MCMC history plots for the parameters of a two compartment model with first order absorption (each color corresponds to a different chain)




FIGURE 4. Posterior Marginal Densities of the Model Parameters of a two compartment model with first order absorption (each color corresponds to a different chain)

where  $K$  is a matrix. For example  $K$  for a two compartment model with first order absorption is:

$$K = \begin{bmatrix} -k_a & 0 & 0 \\ k_a & -(k_{10} + k_{12}) & k_{21} \\ 0 & k_{12} & -k_{21} \end{bmatrix}$$

where  $k_{10} = CL/V2$ ,  $k_{12} = Q/V2$ , and  $k_{21} = Q/V2$ .




FIGURE 5. Predicted (posterior median and 90 % credible intervals) and observed plasma drug concentrations of a two compartment model with first order absorption

The linear compartment model has the form:

```
linCptModel(K, theta, time, amt, rate, ii, evid, cmt, addl, ss)
```

where  $K$  is the matrix describing the ODE system. Since most of the parameters appear in  $K$ , it is not necessary to specify them again in  $\theta$ . Instead, we only specify in  $\theta$  parameters left out of  $K$ , namely the bioavailability fractions and the lag times.

**General Compartment Model.** Torsten may be used to fit models described by a system of first-order ODEs, i.e., differential equations of the form:

$$x'(t) = f(t, x(t))$$

where  $x$  and  $f$  are vector-valued functions.

The general compartment model functions have the form:

```
<modelname>(ODE_system, nCmt,
theta, time, amt, rate, ii, evid, cmt, addl, ss,
reltol, abstol, max_step)
```

where `ODE_system` is a system of first-order ODEs defined in the function block of Stan (see section 19.2 of the Stan reference manual) and `nCmt` is the number of compartments in the model. `reltol`, `abstol`, and `max_step` are the tuning parameters for the ODE integrator, respectively the relative tolerance, the absolute tolerance, and the maximum number of steps. It is difficult to recommend values for the tuning parameters, though the authors typically use  $rel\_tol = 1e - 8$ ,  $abs\_tol = 1e - 8$  and  $max\_step = 1e + 8$ . For more details see Stan's reference manual.

FIGURE 6. Stan language for fitting a two compartment model using the `linCptModel` function (abstract)

```

transformed parameters {
  matrix[3, 3] K;
  real k10;
  real k12;
  real k21;
  vector<lower = 0>[nTheta] theta[1];
  vector<lower = 0>[nt] cHat;
  vector<lower = 0>[nObs] cHatObs;
  matrix<lower = 0>[nt, 3] x;

  k10 = CL / V2;
  k12 = Q / V2;
  k21 = Q / V3;

  K = rep_matrix(0, 3, 3);

  K[1, 1] = -ka;
  K[2, 1] = ka;
  K[2, 2] = -(k10 + k12);
  K[2, 3] = k21;
  K[3, 2] = k12;
  K[3, 3] = -k21;

  theta[1][1] = 1; # F1
  theta[1][2] = 1; # F2
  theta[1][3] = 1; # F3
  theta[1][4] = 0; # tlag1
  theta[1][5] = 0; # tlag2
  theta[1][6] = 0; # tlag3

  x = linCptModel(K, theta, time, amt, rate, ii, evid, cmt, addl, ss);

  cHat = col(x, 2) ./ V1;

  for(i in 1:nObs){
    cHatObs[i] = cHat[iObs[i]]; # predictions for observed data records
  }
}

model{
  logCObs ~ normal(log(cHatObs), sigma);
}

```

The options for `model_name` are:

- `generalCptModel_rk45`
- `generalCptModel_CVODES`

They respectively call the built-in Runge-Kutta 4th/5th integrator, recommended for non-stiff ODEs, and CVODES integrator, recommended for stiff ODEs.

FIGURE 7. Stan language for fitting a two compartment model using the `genCptModel_rk45` function (abstract)

```

functions{
  # define ODE system for two compartment model
  real[] twoCptModelODE(real t,
                        real[] x,
                        real[] theta,
                        real[] rate, # in this example, rate is treated as data
                        int[] dummy){

    real Q;
    real CL;
    real V1;
    real V2;
    real ka;
    real k12;
    real k21;
    real k10;
    real y[3];

    CL = theta[1];
    Q = theta[2];
    V1 = theta[3];
    V2 = theta[4];
    ka = theta[5];
    k10 = CL / V1;
    k12 = Q / V1;
    k21 = Q / V2;

    y[1] = -ka*x[1];
    y[2] = ka*x[1] - (k10 + k12)*x[2] + k21*x[3];
    y[3] = k12*x[2] - k21*x[3];

    return y;
  }
}

:
:
transformed parameters {
:
:

  theta[1][1] = CL;
  theta[1][2] = Q;
  theta[1][3] = V1;
  theta[1][4] = V2;
  theta[1][5] = ka;
  theta[1][6] = 1; # F1
  theta[1][7] = 1; # F2
  theta[1][8] = 1; # F3
  theta[1][9] = 0; # tlag1
  theta[1][10] = 0; # tlag2
  theta[1][11] = 0; # tlag3

  x = generalCptModel_bdf( twoCptModelODE, 3,
                          theta, time, amt, rate, ii, evid, cmt, addl, ss,
                          1e-8, 1e-8, 1e8);
:
:

```

TABLE 2. Arguments of Torsten functions.

model	function name	argument names	model parameters in theta
one compartment model with first order absorption ( $k_a > \frac{CL}{V_2}$ )	PKModelOneCpt	time, amt, rate, ii, evid, cmt, addl, ss	$CL, V_2, k_a, F_1, F_2,$ $t_{lag1}, t_{lag2}$
two compartment model with first order absorption ( $k_a > \lambda_1^a$ )	PKModelTwoCpt	time, amt, rate, ii, evid, cmt, addl, ss	$CL, Q, V_2, V_3, k_a, F_1,$ $F_2, F_3, t_{lag1}, t_{lag2}, t_{lag2}$
general linear compartment model	linCptModel	K, time, amt, rate, ii, evid, cmt, addl, ss	$F_1$ to $F_n, t_{lag1}$ to $t_{lagn}$
general compartment models	genCptModel_*	ODE_system, nCmt, time, amt, rate, ii, evid, cmt, addl, ss, rel_tol, abs_tol, max_num_steps	Parameters in ODE system, $F_1$ to $F_n, t_{lag1}$ to $t_{lagn}$

$$^a \lambda_1 = \frac{k_{10} + k_{12} + k_{21} - \sqrt{(k_{10} + k_{12} + k_{21})^2 - 4k_{10}k_{21}}}{2} \text{ where } k_{10} = \frac{CL}{V_2}, k_{12} = \frac{Q}{V_2} \text{ and } k_{21} = \frac{Q}{V_3}.$$

**Summary.** Table 2 summarizes how to call Torsten functions.

It is key to understand which type of models each function works for, and which method optimizes model fitting. An analytical method will always be fastest, but only applies to a few simple cases. Numerical methods that use an ODE integrator are generally applicable, but are orders of magnitude slower. The matrix exponential solution falls, speed-wise, between the analytical and the numerical solution, and can be used for all linear ODEs system.

## 4. ADDITIONAL EXAMPLES

**Available Examples online.** Code for examples can be found on GitHub: <https://github.com/charlesm93/example-models/tree/feature/issue-70-PKPDexamples-torsten/PKPD/torsten>.

The current version of Torsten requires a development version of Stan that has not yet been released. In order to run this version, we advice users to use cmdStan, the command line interface of Stan. When Stan v2.13 (expected end of November 2016) gets released, we will share examples that use RStan.

All the files to run a model are stored under the directory that bears the model's name. There are four files per example:

- <model name>.stan
- <model name>.data.R
- <model name>.init.R
- <model name>Simulation.R

data.R contains the data we fit the model to and init.R the initial estimates of the parameters. These two files are generated using Simulation.R. In order to run the model, I recommend creating a new directory under cmdstan/examples, next to the *bernouilli* example Stan automatically installs. Under this directory, put the data.R, init.R, and .stan files. You can then compile the model using:

```
make examples/<model name>/<model name>
```

and run it:

```
cd examples/<model name>
./<model name> sample data file=<model name>.data.R init=<model name>.init.R
```

**Example 2: Effect Compartment Model.** Let us expand example 1 to a population model fitted to the combined data from phase I and phase IIa studies. The parameters exhibit inter-individual variations (IIV), due to both random effects and to the patients' body weight, treated as a covariate and denoted *bw*:

*Population Model for Plasma Drug Concentration (c).*

$$\begin{aligned}
 \log(c_{ij}) &\sim N(\log(\hat{c}_{ij}), \sigma^2) \\
 \hat{c}_{ij} &= f_{2cpt}(t_{ij}, D_j, \tau_j, CL_j, Q_j, V_{1j}, V_{2j}, k_{aj}) \\
 \log(CL_j, Q_j, V_{ssj}, k_{aj}) &\sim N\left(\log\left(\widehat{CL}\left(\frac{bw_j}{70}\right)^{0.75}, \widehat{Q}\left(\frac{bw_j}{70}\right)^{0.75}, \widehat{V}_{ss}\left(\frac{bw_j}{70}\right), \widehat{k}_a\right), \Omega\right) \\
 V_{1j} &= f_{V_1} V_{ssj} \quad V_{2j} = (1 - f_{V_1}) V_{ssj} \\
 (\widehat{CL}, \widehat{Q}, \widehat{V}_{ss}, \widehat{k}_a, f_{V_1}) &= (10 \text{ L/h}, 15 \text{ L/h}, 140 \text{ L}, 2 \text{ h}^{-1}, 0.25) \\
 \Omega &= \begin{pmatrix} 0.25^2 & 0 & 0 & 0 \\ 0 & 0.25^2 & 0 & 0 \\ 0 & 0 & 0.25^2 & 0 \\ 0 & 0 & 0 & 0.25^2 \end{pmatrix}, \quad \sigma = 0.1
 \end{aligned}$$

Furthermore we add a fourth compartment in which we measure a PD effect (figure 8).




FIGURE 8. Effect Compartment Model

*Effect Compartment Model for PD response (R).*

$$\begin{aligned}
 R_{ij} &\sim N\left(\hat{R}_{ij}, \sigma_R^2\right) \\
 \hat{R}_{ij} &= \frac{E_{max} c_{eij}}{EC_{50j} + c_{eij}} \\
 c'_{e,j} &= k_{e0j} (c_{.j} - c_{e,j}) \\
 \log(EC_{50j}, k_{e0j}) &\sim N\left(\log\left(\widehat{EC}_{50}, \widehat{k}_{e0}\right), \Omega_R\right) \\
 \left(E_{max}, \widehat{EC}_{50}, \widehat{k}_{e0}\right) &= (100, 100.7, 1) \\
 \Omega_R &= \begin{pmatrix} 0.2^2 & 0 \\ 0 & 0.25^2 \end{pmatrix}, \quad \sigma_R = 10
 \end{aligned}$$

The PK and the PD data are simulated using the following treatment.

- Phase I study
  - Single dose and multiple doses
  - Parallel dose escalation design
  - 25 subjects per dose
  - Single doses: 1.25, 5, 10, 20, and 40 mg
  - PK: plasma concentration of parent drug ( $c$ )
  - PD response: Emax function of effect compartment concentration ( $R$ )
  - PK and PD measured at 0.083, 0.167, 0.25, 0.5, 0.75, 1, 2, 3, 4, 6, 8, 12, 18, and 24 hours

- Phase IIa trial in patients
  - 100 subjects
  - Multiple doses: 20 mg
  - sparse PK and PD data (3-6 samples per patient)

The model is now simultaneously fitted to the PK and the PD data! For this four compartment model, we construct a constant rate matrix and use `linCptModel`. Correct use of `Torsten` requires that the user pass the entire event history (observation and dosing events) for an individual to the function. Thus the Stan model shows the call to `linCptModel` within a loop over the individual subjects rather than over the individual observations.

*Results.* We use the same diagnosis tools as for the previous example. The MCMC history plots (figure 10) suggest the 4 chains have converge to common distributions. We note some minor auto-correlations for  $lp_{-}$  (the log posterior) and for IIV parameters: specifically  $\Omega_{ke,0}$  and  $\rho$ . The correlation matrix  $\rho$  does not explicitly appear in the model, but it is used to construct  $\Omega$ , which parametrizes the PK IIV. The fits to the plasma concentration (figure 12) are in close agreement with the data, notably for the sparse data case (phase IIa study). The fits to the PD data (figure 13) look good, though the data is more noisy. The model reflects the noise by producing larger confidence intervals. The estimated values of the parameters are consistent with the values used to simulate the data.

TABLE 3. Summary of the MCMC simulations of the marginal posterior distributions of the model parameters for example 2

	mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
CLHat	10.095	0.003	0.201	9.712	9.958	10.096	10.231	10.483	4000.000	0.999
QHat	14.867	0.014	0.357	14.182	14.620	14.862	15.106	15.563	678.208	1.007
V1Hat	34.188	0.067	1.089	31.940	33.494	34.214	34.918	36.251	267.748	1.016
V2Hat	103.562	0.076	2.925	98.031	101.600	103.455	105.472	109.583	488.296	1.001
kaHat	1.930	0.004	0.077	1.771	1.880	1.933	1.982	2.076	334.888	1.014
ke0Hat	1.050	0.001	0.044	0.967	1.020	1.051	1.078	1.137	164.741	1.000
EC50Hat	104.337	0.040	2.100	100.169	102.909	104.345	105.768	108.351	744.041	1.000
sigma	0.099	0.000	0.002	0.095	0.097	0.099	0.100	0.103	906.342	1.002
sigmaResp	10.156	0.003	0.197	9.779	10.023	10.154	10.286	10.552	4000.000	1.000
omega[1]	0.270	0.000	0.016	0.241	0.259	0.269	0.280	0.302	4000.000	1.001
omega[2]	0.231	0.001	0.021	0.192	0.217	0.230	0.245	0.275	531.512	1.006
omega[3]	0.219	0.002	0.031	0.158	0.199	0.218	0.238	0.281	158.198	1.017
omega[4]	0.267	0.001	0.026	0.218	0.249	0.266	0.284	0.319	684.870	1.001
omega[5]	0.285	0.002	0.037	0.214	0.259	0.284	0.309	0.361	284.545	1.009
omegaKe0	0.271	0.003	0.047	0.183	0.239	0.271	0.303	0.363	217.350	1.007
omegaEC50	0.213	0.001	0.021	0.174	0.199	0.213	0.227	0.255	190.193	1.000

**Example 3: Friberg-Karlsson Semi-Mechanistic Model [3].** In this third example, we deal with a more sophisticated PD effect, described by a system of nonlinear ODEs. The PK effects are still described by a two compartment model with a first-order absorption.

Neutropenia is observed in patients receiving an ME-2 drug. Our goal is to model the relation between neutrophil counts and drug exposure. Using a feedback mechanism, the body maintains the number of neutrophils at a baseline value (figure 14). While in the patient’s blood, the drug impedes the production of neutrophils. As a result, the neutrophil count goes down, and after the drug clears out, the feedback mechanism kicks in and brings the neutrophil count back to baseline.



FIGURE 9. Stan language for fitting an effect compartment model using `linCptModel` (abstract)

```

transformed parameters {
  for(j in 1:nSubjects){
    :
    :
    Omega = quad_form_diag(rho, omega);

    for(j in 1:nSubjects){
      CL[j] = exp(logtheta[j, 1]) * (weight[j] / 70)^0.75;
      Q[j] = exp(logtheta[j, 2]) * (weight[j] / 70)^0.75;
      V1[j] = exp(logtheta[j, 3]) * weight[j] / 70;
      V2[j] = exp(logtheta[j, 4]) * weight[j] / 70;
      ka[j] = exp(logtheta[j, 5]);
      ke0[j] = exp(logKe0[j]);
      EC50[j] = exp(logEC50[j]);

      k10 = CL[j] / V1[j];
      k12 = Q[j] / V1[j];
      k21 = Q[j] / V2[j];

      K = rep_matrix(0, 4, 4);
      K[1, 1] = -ka[j];
      K[2, 1] = ka[j];
      K[2, 2] = -(k10 + k12);
      K[2, 3] = k21;
      K[3, 2] = k12;
      K[3, 3] = -k21;
      K[4, 2] = ke0[j];
      K[4, 4] = -ke0[j];

      ke0[j] = exp(logKe0[j]);
      EC50[j] = exp(logEC50[j]);

      k10 = CL[j] / V1[j];
      k12 = Q[j] / V1[j];
      k21 = Q[j] / V2[j];

      K = rep_matrix(0, 4, 4);

      K[1, 1] = -ka[j];
      K[2, 1] = ka[j];
      K[2, 2] = -(k10 + k12);
      K[2, 3] = k21;
      K[3, 2] = k12;
      K[3, 3] = -k21;
      K[4, 2] = ke0[j];
      K[4, 4] = -ke0[j];

      x[start[j]:end[j],] = linCptModel(K, parms, time[start[j]:end[j]],
                                      amt[start[j]:end[j]], rate[start[j]:end[j]],
                                      ii[start[j]:end[j]], evid[start[j]:end[j]],
                                      cmt[start[j]:end[j]], addl[start[j]:end[j]],
                                      ss[start[j]:end[j]]);

      cHat[start[j]:end[j]] = 1000 * x[start[j]:end[j], 2] ./ V1[j];
      ceHat[start[j]:end[j]] = 1000 * x[start[j]:end[j], 4] ./ V1[j];
      respHat[start[j]:end[j]] = 100 * ceHat[start[j]:end[j]] ./
        (EC50[j] + ceHat[start[j]:end[j]]);
    }

    for(i in 1:nObs){
      cHatObs[i] = cHat[iObs[i]];
      respHatObs[i] = respHat[iObs[i]];
    }
  }
  :
  :
}

```




FIGURE 10. MCMC history plots for the parameters of an Effect Compartment Model (each color corresponds to a different chain) for example 2




FIGURE 11. Posterior Marginal Densities of the Model Parameters of an Effect Compartment Model (each color corresponds to a different chain) for example 2




FIGURE 12. Predicted (posterior median and 90 % credible intervals) and observed plasma drug concentrations for example 2 for an Effect Compartment Model

### Friberg-Karlsson Model for drug-induced myelosuppression ( $ANC$ ).

$$\begin{aligned} \log(ANC_{ij}) &\sim N(Circ_{ij}, \sigma_{ANC}^2) \\ \log(MTT_j, Circ_{0j}, \alpha_j) &\sim N\left(\log\left(\widehat{MTT}, \widehat{Circ_0}, \widehat{\alpha}\right), \Omega_{ANC}\right) \\ \left(\widehat{MTT}, \widehat{Circ_0}, \widehat{\alpha}, \gamma\right) &= (125, 5, 2, 0.17) \end{aligned}$$

$$\Omega_{ANC} = \begin{pmatrix} 0.2^2 & 0 & 0 \\ 0 & 0.35^2 & 0 \\ 0 & 0 & 0.2^2 \end{pmatrix}, \quad \sigma_{ANC} = 0.1$$

$$\Omega_{PK} = \begin{pmatrix} 0.25^2 & 0 & a0 & 0 & 0 \\ 0 & 0.4^2 & 0 & 0 & 0 \\ 0 & 0 & 0.25^2 & 0 & 0 \\ 0 & 0 & 0 & 0.4^2 & 0 \\ 0 & 0 & 0 & 0 & 0.25^2 \end{pmatrix}$$



FIGURE 13. Predicted (posterior median and 90 % credible intervals) and observed PD Response for example 2




FIGURE 14. Friberg-Karlsson semi-mechanistic Model [3]

The PK and the PD data are simulated using the following treatment.

- Phase IIa trial in patients
  - Multiple doses: 80,000 mg
  - Parallel dose escalation design
  - 15 subjects
  - PK: plasma concentration of parent drug ( $c$ )
  - PD response: Neutrophil count ( $ANC$ )
  - PK measured at 0.083, 0.167, 0.25, 0.5, 0.75, 1, 2, 3, 4, 6, 8, 12, 18, and 24 hours
  - PD measured once every two days for 28 days.

Once again, we simultaneously fit the model to the PK and the PD data. From a computational perspective, this is a much more difficult problem than the one we dealt with in previous examples. The nonlinear nature of the ODEs forces us to use a numerical solver, which is significantly slower than the linear methods we have employed so far. Because the ODE system of interest is non-stiff, we use the *rk45* version of `genCptModel`.

It pays off to construct informative priors. For instance, we could fit the PK data first, as was done in example 1, and get informative priors on the PK parameters. The PD parameters are drug independent,

FIGURE 15. Stan language for coding an ODE system describing a Friberg-Karlsson Mechanism

```

real[] twoCptNeutModelODE(real t,
                          real[] x,
                          real[] parms,
                          real[] rdummy,
                          int[] idummy){

  real k10;
  real k12;
  real k21;
  real CL;
  real Q;
  real V1;
  real V2;
  real ka;
  real mtt;
  real circ0;
  real gamma;
  real alpha;
  real ktr;
  real dxdt[8];
  real conc;
  real EDrug;
  real transit1;
  real transit2;
  real transit3;
  real circ;
  real prol;

  CL = parms[1];
  Q = parms[2];
  V1 = parms[3];
  V2 = parms[4];
  ka = parms[5];
  mtt = parms[6];
  circ0 = parms[7];
  gamma = parms[8];
  alpha = parms[9];

  k10 = CL / V1;
  k12 = Q / V1;
  k21 = Q / V2;

  ktr = 4 / mtt;

  dxdt[1] = -ka * x[1];
  dxdt[2] = ka * x[1] - (k10 + k12) * x[2] + k21 * x[3];
  dxdt[3] = k12 * x[2] - k21 * x[3];
  conc = x[2]/V1;
  EDrug = alpha * conc;
  // x[4], x[5], x[6], x[7] and x[8] are differences from circ0.
  prol = x[4] + circ0;
  transit1 = x[5] + circ0;
  transit2 = x[6] + circ0;
  transit3 = x[7] + circ0;
  circ = fmax(machine_precision(), x[8] + circ0); // Device for implementing a modeled
                                                    // initial condition
  dxdt[4] = ktr * prol * ((1 - EDrug) * ((circ0 / circ)^gamma) - 1);
  dxdt[5] = ktr * (prol - transit1);
  dxdt[6] = ktr * (transit1 - transit2);
  dxdt[7] = ktr * (transit2 - transit3);
  dxdt[8] = ktr * (transit3 - circ);

  return dxdt;
}

```

so we can use information from the neutropenia literature. In this example, we choose to use weakly informative priors on the PK parameters and strongly informative priors on the PD parameters.

Since it takes a long time to run the model, we only use 100 iterations per chain, and study what we can learn from this less than optimal scenario. It is worth noting that Stan, because of its highly efficient MCMC sampler, still does a reasonable job estimating the posterior distribution.

FIGURE 16. Stan language for fitting a Friberg-Karlsson model using `genCptModel_rk45` (abstract)

```

transformed parameters {
  :
  for(i in 1:nSubjects) {

    parms[1][1] = thetaM[i, 1] * (weight[i] / 70)^0.75; # CL
    parms[1][2] = thetaM[i, 2] * (weight[i] / 70)^0.75; # Q
    parms[1][3] = thetaM[i, 3] * (weight[i] / 70); # V1
    parms[1][4] = thetaM[i, 4] * (weight[i] / 70); # V2
    parms[1][5] = kaHat; # ka
    parms[1][6] = thetaM[i, 5]; # mtt
    parms[1][7] = thetaM[i, 6]; # circ0
    parms[1][8] = gamma;
    parms[1][9] = thetaM[i, 7]; # alpha

    x[start[i]:end[i]] = generalCptModel_rk45(twoCptNeutModelODE, 8,
                                              parms,
                                              time[start[i]:end[i]],
                                              amt[start[i]:end[i]],
                                              rate[start[i]:end[i]],
                                              ii[start[i]:end[i]],
                                              evid[start[i]:end[i]],
                                              cmt[start[i]:end[i]],
                                              addl[start[i]:end[i]],
                                              ss[start[i]:end[i]],
                                              1e-6, 1e-6, 1e8);

    cHat[start[i]:end[i]] = x[start[i]:end[i], 2] / parms[1][3]; # divide by V1
    neutHat[start[i]:end[i]] = x[start[i]:end[i], 8] + parms[1][7]; # Add baseline
  }

  for(i in 1:nObsPK) cHatObs[i] = cHat[iObsPK[i]];
  for(i in 1:nObsPD) neutHatObs[i] = neutHat[iObsPD[i]];

  :
}

```

*Results.* The MCMC history plots are not as convincing as in the previous examples, mostly because the number of iterations is small (100 versus 1000 in the previous example). It does however look as though the chains are converging to a common distribution, and we see little auto-correlation (in particular, we expect that if we had run the model for 1000 iterations, we would obtain the desired "fuzzy caterpillar" look). The plots of the marginal posterior distributions clearly show that the chains have not (yet) converged to a common distribution, but they do not disagree significantly. Still, the need for more iterations is evident. The model fits the data, and the confidence interval reflect the noise in the data. The parameters estimation are actually pretty good.




FIGURE 17. MCMC history plots for the parameters of a Friberg-Karlsson semi-mechanistic model (each color corresponds to a different chain) for example 3




FIGURE 18. Posterior Marginal Densities of the Model Parameters of a Friberg-Karlsson semi-mechanistic model (each color corresponds to a different chain)




FIGURE 19. Predicted (posterior median and 90 % credible intervals) and observed plasma drug concentrations, and Neutrophil counts, for a Friberg-Karlsson semi-mechanistic model

TABLE 4. Summary of the MCMC simulations of the marginal posterior distributions of the model parameters for example 3

	mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
CL	9.986	0.009	0.174	9.641	9.872	9.982	10.107	10.331	400.000	0.997
Q	14.633	0.055	1.106	12.505	13.992	14.623	15.296	16.948	400.000	0.996
V1	32.909	0.174	2.439	28.203	31.186	32.836	34.762	37.750	195.828	1.008
V2	106.631	0.311	6.226	95.234	102.269	106.403	111.000	118.533	400.000	0.999
ka	1.882	0.012	0.175	1.582	1.756	1.871	2.006	2.223	196.052	1.007
sigma	0.106	0.001	0.010	0.089	0.098	0.105	0.112	0.132	259.693	1.009
alpha	3.3E-04	1.4E-06	2.2E-05	2.9E-04	3.2E-04	3.3E-04	3.5E-04	3.8E-04	247	1.01
mtt	132.763	0.515	6.498	120.843	128.082	132.223	136.694	146.845	159.372	1.024
circ0	5.014	0.009	0.172	4.711	4.888	5.000	5.138	5.334	400.000	1.000
gamma	0.190	0.002	0.022	0.153	0.175	0.187	0.202	0.239	139.485	1.025
sigmaNeut	0.092	0.001	0.014	0.068	0.082	0.090	0.100	0.125	161.199	1.010



## 5. UNDER THE HOOD DESIGN

We here discuss some background theory and the design of Torsten at a C++ level.

Our approach is heavily based on the *BUGS model library*, a prototype PKPD model library for WinBUGS (<https://bitbucket.org/metrumrg/bugsmodellibrary/wiki/Home>), developed by Metrum Research Group in 2009.

This section is intended for developers. No knowledge of C++ is required for users.

**Computing Amounts in an ODE-based model.** Most PKPD model are based on ODEs that describe how PK and PD amounts evolve over time. For instance, the following ODEs describe drug diffusion in a one compartment model with a first-order absorption from the gut:

$$\begin{aligned}\frac{dGUT}{dt} &= -kaGUT \\ \frac{dCENT}{dt} &= kaGUT - \frac{CL}{V_1}CENT\end{aligned}$$

If the ODEs fully describe the PKPD system, knowing the state  $y_0$  at time  $t_0$  fully defines the solution at finite times. Exploiting this property, Torsten calculates the evolution of amounts in each compartment from one event to the other. The initial conditions of the ODE system are specified by the previous event, and the ODEs are integrated from  $t_{previous}$  to  $t_{current}$ .

Note we cannot simply integrate from  $t_{first}$  to  $t_{last}$  because the ODEs do not describe exterior interventions, such as additional dosing. Torsten treats these interventions independently. Most importantly, Torsten only integrates between  $t_0$  and  $t_1$  if no exterior interventions occur during this interval. To achieve this, it is key to properly handle the *event schedule*.

All five functions in Torsten call the C++ function `pred`, which:

- (1) augments the event schedule to include all events that alter the system
- (2) calculates the amounts in each compartment at each event of the augmented schedule
  - computes the *natural* evolution of the system by integrating ODEs
  - computes alterations due to exterior interventions
- (3) returns the amounts at each event of the original schedule

The Event Schedule depends on the user's input (TIME, EVID, CMT, AMT, RATE, ADDL, II, SS). The event schedule may need to be augmented if, for example, an event specifies a patient receives multiple doses at a regular time interval. Consider:

```
TIME = 0, EVID = 1, CMT = 1, AMT = 1500, RATE = 0, ADDL = 4, II = 10, SS = 0
```

This Event specifies that at time 0 (TIME = 0), a patient receives a 1500 mg (AMT = 1500) drug dose (EVID = 1) in the gut (CMT = 1), and will receive an additional dose every 10 hours (II = 10) until the patient has received a total of 5 doses (ADDL = 4, being the number of additional doses, + 1, the original dose). Such an Event really corresponds to 5 dosing events.

To integrate the ODEs, `pred` calls `pred1` (prediction for one event) or `predSS` (prediction for one event if the system is in a steady state, i.e  $SS = 1$ ). `pred1` and `predSS` are functors and get constructed differently by each Torsten function. Under `PKModelOneCpt`, `pred1` analytically computes the solution, while under `generalCptModel_rk45` it numerically solves the ODEs.

**Structure of a Torsten Function.** Under the structural scheme described above, a Torsten function performs a very simple set of actions:

- (1) Consistent with Stan practices, check the validity of the arguments and of the parameter values
- (2) Construct the `PKModel` object, which contains basic information about the model, such as the number of compartments
- (3) Construct the `pred1` and `predSS` functions
- (4) Call `pred`

Figure 20 shows the C++ code for `PKModelOneCpt`.

### Implementing Torsten in Stan.

*Modifications in Stan-math.* All five Torsten functions are located under the `Torsten` directory, under `stan/math`. We modified `rev/math` to include the `torsten/torsten.hpp` header file. The code can be found on GitHub: <https://github.com/charlesm93/math>

*Modifications in Stan.* Further modification are done in Stan to expose the Torsten functions to the Stan language. We edited `function_signatures.h` to expose `PKModelOneCpt`, `PKModelTwoCpt`, and `linCptModel`. The general compartment model functions are higher-order functions in that they take another function as one of their arguments. They were exposed by directly modifying the grammar files, following very closely the example of `integrate_ode_rk45` and `integrate_ode_bdf`.

The code can be found on GitHub: <https://github.com/charlesm93/stan>

FIGURE 20. C++ code for the PKModelOneCpt function (abstract)

```

template <typename T0, typename T1, typename T2, typename T3, typename T4>
Eigen::Matrix <typename promote_args<T0, T1, T2, T3, T4>::type, Eigen::Dynamic,
  Eigen::Dynamic>
PKModelOneCpt(const std::vector< Eigen::Matrix<T0, Eigen::Dynamic, 1> >& pMatrix,
               const std::vector<T1>& time,
               const std::vector<T2>& amt,
               const std::vector<T3>& rate,
               const std::vector<T4>& ii,
               const std::vector<int>& evid,
               const std::vector<int>& cmt,
               const std::vector<int>& addl,
               const std::vector<int>& ss) {

  using std::vector;
  using Eigen::Dynamic;
  using Eigen::Matrix;
  using boost::math::tools::promote_args;
  using stan::math::check_positive_finite;

  PKModel model("OneCptModel");

  // Check arguments
  static const char* function("PKModelOneCpt");
  pmetricsCheck(pMatrix, time, amt, rate, ii, evid, cmt, addl, ss, function, model);
  for(int i=0; i<pMatrix.size(); i++) {
    check_positive_finite(function, "PK parameter CL", pMatrix[i](0,0));
    check_positive_finite(function, "PK parameter V2", pMatrix[i](1,0));
    check_positive_finite(function, "PK parameter ka", pMatrix[i](2,0));
  }
  std::string message4 = ", but must equal the number of parameters in the model: "
    + boost::lexical_cast<string>(model.GetNParameter()) + "!";
  const char* length_error4 = message4.c_str();
  if (!(pMatrix[0].size() == model.GetNParameter()))
    stan::math::invalid_argument(function,
      "The number of parameters per event (length of a vector in the first argument) is",
      pMatrix[0].size(), "", length_error4);

  // Construct Pred functions for the model.
  Pred1_structure new_Pred1("OneCptModel");
  PredSS_structure new_PredSS("OneCptModel");
  Pred1 = new_Pred1;
  PredSS = new_PredSS;

  // Construct dummy matrix for last argument of pred
  Eigen::Matrix<double, Dynamic, Dynamic> dummy_system(0,0);

  Matrix <typename promote_args<typename promote_args<T0, T1, T2, T3, T4>::type,
    double>::type, Dynamic, Dynamic> pred;
  pred = Pred(pMatrix, time, amt, rate, ii, evid, cmt, addl, ss, model,
    dummy_ode(), dummy_system);

  return pred;
}

```

## REFERENCES

- [1] Carpenter, B., Gelman, A., Hoffman, M., Lee, D., Goodrich, B., Betancourt, M., Brubaker, M.A., Guo, J., Li, P. and Riddell, A. Stan: A probabilistic programming language. *Journal of Statistical Software (in press)* (2016).
- [2] Moler, C. and Van Loan, C. Nineteen dubious ways to compute the exponential of a matrix, twenty-five years later. *SIAM Review* (2003).
- [3] Friberg, L.E. and Karlsson, M.O. Mechanistic models for myelosuppression. *Invest New Drugs* **21** (2003):183–194.